

Ad, Soyad:	Hava Harp Okulu	S1	S2	S3	Σ
Numara:	Bilgisayar Mühendisliği Bölümü	(20p)	(40p)	(40p)	(100)
Tarih:	Mikroişlemciler ve Mikrobilgisayarlar				
	Vize 1				

Sınavda öğrenci **kendisine ait** a) komut kümesi ve b) EXE tipi kod iskeleti fotokopilerini kullanabilir.

Başarılar.

**Soru 1) Boşluk doldurma ve Doğru-Yanlış soruları: (20p)**

- a) Aşağıdaki boşlukları uygun şekilde doldurunuz. **Sadece 6 adet şıkkı cevaplayınız. (12p)**
- 8086 mikroişlemcisinin 20 adet adres ucu vardır.
  - 8086 mikroişlemcisi, verinin düşük anlamlı byte değerini küçük adreste saklaması özelliği ile little endian formatındadır.
  - Tek operandı olan assembly komutlarına örnek olarak MUL verilebilir.
  - Operandı olmayan assembly komutlarına örnek olarak NOP verilebilir.
  - SS** 16 bitlik bir yazmaçtır.
  - AH** 8 bitlik bir yazmaçtır.
  - Sekiz bitlik **10001000** binary sayısı işaretli sayı olarak değerlendirildiğinde onluk tabandaki karşılığı -120 olur.
  - IP** ile görelili adresleme için varsayılan kesim yazmacı CS 'dır.

- b) Aşağıda verilen ifadelerin başına **D** (doğru) veya **Y** (yanlış) olduklarını işaretleyiniz. **Sadece 4 adet şıkkı cevaplayınız. (8p)**

D	Doğrudan üç operandı olan 8086 assembly komutu yoktur.
D	<b>MOV</b> komutunun yürütülmesinden sonra bayraklar etkilenmez.
Y	<b>LEA SI, data</b> komutu ile <b>MOV SI, data</b> komutunu yürütülmesi <b>SI</b> 'da her zaman aynı sonucu üretir.
Y	<b>CMP</b> komutu operandları arasında bitwise AND işlemi uygular, sonuç saklanmaz sadece bayraklar uygun şekilde değiştirilir.
Y	8086 mikroişlemcisinin 8 adet veri ucu mevcuttur.
D	<b>NOT AX</b> komutu ile <b>XOR AX, 0FFFFH</b> komutunun yürütülmesi <b>AX</b> 'de her zaman aynı sonucu üretir.

Soru 2) Aşağıdaki komut blokları yürütüldükten sonra belirtilen yazmacın alacağı değeri yazınız. Yaptığınız analiz ve hesaplamaları açıkça yazınız. (40p)

a)

```
MOV AH, 1
MOV AL, 11111111B
CMP AL, 2
JG L1
MOV AH, 2
L1:
```

, AH=? (10p)

JG işaretli sayıların karşılaştırılmasında kullanılır 11111111B işaretli sayı olarak ele alınırsa -1'e eşittir -1 ile 2 sayıları CMP ile karşılaştırıldığında JG'den false gelir

AH=2 olur

b)

```
MOV AH, 10
MOV BH, 10H
CMP AH, BH
JA L1
XCHG AH, BH
L1:
```

, AH=? (10p)

AH ve BH yazmaçlarındaki değerlerden işaretsiz sayı olarak büyük olanı AH'a küçük olanını BH'a aktarılır AH=0AH ve BH=10H ile ilklendirilmiştir

AH=10H olur

c)

```
MOV AL, 20
MOV BL, 12
L1: CMP BL, 0
JZ L2
XOR AH, AH
DIV BL
MOV AL, BL
MOV BL, AH
JMP L1
L2:
```

, AH=? (10p)

Öklid yöntemi ile OBEB hesaplar, (A>B iken)

$$GCD(A, B) = \begin{cases} A & B = 0 \\ GCD(B, A \% B) & B \neq 0 \end{cases}$$

İşlem	İşlem Sonrası yazmaçlar (Hex)			
	AX	AH	AL	BL
MOV AL, 20	?	?	14	?
MOV BL, 12	?	?	14	0C
CMP BL, 0 JZ L2	false			
XOR AH, AH	0014	00	14	0C
DIV BL	0801	08	01	0C
MOV AL, BL	080C	08	0C	0C
MOV BL, AH	080C	08	0C	08
CMP BL, 0 JZ L2	false			
XOR AH, AH	000C	00	0C	08
DIV BL	0401	04	01	08
MOV AL, BL	0408	04	08	08
MOV BL, AH	0408	04	08	04
CMP BL, 0 JZ L2	false			
XOR AH, AH	0008	00	08	04
DIV BL	0002	00	02	04
MOV AL, BL	0004	00	04	04
MOV BL, AH	0004	00	04	00
CMP BL, 0 JZ L2	true			

AH=0, AL=4

```

XOR DX, DX
MOV AX, 25
MOV BX, 5
L3: CMP BL, 0
JZ L1
TEST BL, 1
JZ L2
ADD DX, AX
L2: SHR BL, 1
SHL AX, 1
JMP L3
L1:

```

d)

, DX=? (10p)

İki sayının çarpımını hesaplar

İşlem	İşlem Sonrası yazmaçlar (Hex)		
	DX	AX	BL
XOR DX, DX	0000	?	?
MOV AX, 25	0000	0019	?
MOV BX, 5	0000	0019	05
CMP BL, 0 JZ L1	false		
TEST BL, 1 JZ L2	false		
ADD DX, AX	0019	0019	05
SHR BL, 1	0019	0019	02
SHL AX, 1	0019	0032	02
CMP BL, 0 JZ L1	false		
TEST BL, 1 JZ L2	true		
SHR BL, 1	0019	0032	01
SHL AX, 1	0019	0064	01
CMP BL, 0 JZ L1	false		
TEST BL, 1 JZ L2	false		
ADD DX, AX	007D	0064	01
SHR BL, 1	007D	0064	00
SHL AX, 1	007D	00C8	00
CMP BL, 0 JZ L1	true		

DX=125

**Soru 3)** Veri kesiminde **byte** tipinde **GIRIS** isimli **500** elemanlı bir dizi tanımlanmıştır. Bu dizinin elemanları **sadece 0-9 arasında değerler alabilmektedir**. Dizi içerisinde 0-9 arasında değerlerden kaçar tane olduğunun **SAYAC** isimli 10 elemanlı diziye yazılması istenmektedir (SAYAC dizisinin ilk gözüne GIRIS dizisindeki 0'ların sayısı, SAYAC dizisinin ikinci gözüne GIRIS dizisindeki 1'lerin sayısı yazılmalı, diğer SAYAC gözleri de benzer şekilde doldurulmalıdır). Buna göre verilen problem için uygun bir veri kesimi tanımlı yaparak, GIRIS isimli dizide her değerden kaçar tane olduğunu SAYAC isimli diziye yazan EXE tipindeki assembly kodunu yazınız. (GIRIS dizisi için: **GIRIS DB 500 DUP(?)** tanımını kullanınız) **(40p)**

```
STACKSG SEGMENT PARA STACK 'STACK'
    DW 32 DUP(?)
STACKSG ENDS

DATASG SEGMENT PARA 'DATA'
GIRIS DB 0,1,2,490 DUP(0),3,4,5,6,7,8,9
ELEMAN DW 500
SAYAC DW 10 DUP(0) ;SAYAC word tipinde tanımlanmalı
DATASG ENDS

CODESG SEGMENT PARA 'CODE'
    ASSUME CS:CODESG, DS:DATASG, SS:STACKSG

BASLA PROC FAR
    PUSH DS
    XOR AX,AX
    PUSH AX

    MOV AX, DATASG
    MOV DS, AX

    LEA SI, GIRIS ;GIRIS dizisinin adresi SI'da
    LEA DI, SAYAC ;SAYAC dizisinin adresi DI'da
    MOV CX, ELEMAN ;dongu sayısı CX'te
    XOR BX, BX ;BX, BH ve BL sıfırlanıyor
L1:
    MOV BL, [SI] ;GIRIS dizisinin sıradaki
                ;elemanı okunuyor

    SHL BL, 1
    INC WORD PTR [DI+BX] ;SAYAC dizisinin arttırılacak
                        ;adresini DI+2x[SI]
    INC SI ;sonraki GIRIS elemanı için
    LOOP L1

    RETF
BASLA ENDP
CODESG ENDS
END BASLA
```